

(c) Silvaco Inc., 2024

3D TFT simulation

Required Modules

Victory Device

-3D Simulator

-3D TFT

Example Name

SET name="TFT_ex02"

DEVICE SIMULATION

GO victorydevice simflags="-P 4"

STRUCTURE SPECIFICATION

MESH three.d

X.MESH location=0 spacing=0.5

X.MESH location=2 spacing=0.5

X.MESH location=3.5 spacing=1

X.MESH location=5 spacing=0.25

X.MESH location=10 spacing=1.5

X.MESH location=15 spacing=0.25

X.MESH location=16.5 spac=1

X.MESH location=18 spacing=0.5

X.MESH location=20 spacing=0.5

Y.MESH location=-0.2 spacing=0.05

Y.MESH location=-0.1 spacing=0.05

Y.MESH location=0 spacing=0.0075

Y.MESH location=0.05 spacing=0.01

Y.MESH location=0.1 spacing=0.0075

Y.MESH location=0.5 spacing=0.25

#Y.MESH location=10 spacing=5

Z.MESH location=0 spacing=5

Z.MESH location=15 spacing=5

X.MESH, Y.MESH, Z.MESH: 각 축의 좌표(location)와 그 지점에서의 격자 간격(spacing)을 정의한다. 전류가 급격히 변하거나 전계(Electric Field)가 집중되는 채널 영역(X=5~15)과

Gate 산화막 경계면($Y=0$ 부근)은 spacing을 매우 작게(0.25, 0.0075 등) 설정하여 해상도를 높였다. 반면, 소자 끝부분은 격자를 크게 잡아 연산 효율을 높였다.

```
# Regions
# 1=oxide 2=silicon 3=oxide
REGION number=1 y.maximum=0 oxide
REGION number=2 y.minimum=0 y.maximum=0.1 silicon
REGION number=3 y.minimum=0.1 oxide
```

REGION: 공간에 물질을 채운다.

1번 층 (위쪽): $Y < 0$ 구간은 Oxide이다. 전기가 통하지 않는 절연체이다.

2번 층 (중간): $Y = 0 \sim 0.1$ 구간은 Silicon이다. 이 장치의 주인공으로, 나중에 전류가 흐르는 통로가 된다.

3번 층 (아래쪽): $Y > 0.1$ 구간은 다시 Oxide(산화막)를 깔아 바닥을 만든다.

```
# Electrodes
# 1=gate 2=source 3=drain 4=substrate
ELECTRODE number=1 name=gate \
  x.minimum=5 x.maximum=15 y.minimum=-0.2 y.maximum=-0.1
ELECTRODE number=2 name=source \
  x.minimum=0 x.maximum=2 y.minimum=-0.1 y.maximum=0
ELECTRODE number=3 name=drain \
  x.minimum=18 x.maximum=20 y.minimum=-0.1 y.maximum=0
ELECTRODE number=4 name=substrate substrate
```

Gate: 위치 $X=5\sim15$, $Y=-0.2\sim-0.1$. 소자의 정중앙 위에 떠 있다. 여기에 전압을 걸면 실리콘 층에 전류가 흐를지 말지 결정하는 "수도꼭지" 역할을 한다.

Source: 위치 $X=0\sim2$, $Y=-0.1\sim0$. 전자가 들어오는 입구이다.

Drain: 위치 $X=18\sim20$, $Y=-0.1\sim0$. 전자가 나가는 출구이다.

```
# Doping profiles
DOPING region=2 uniform concentration=7e14 n.type
DOPING region=2 gauss concentration=3e18 n.type x.right=5 character=0.3
DOPING region=2 gauss concentration=3e18 n.type x.left=15 character=0.3
```

DOPING: 반도체에 불순물을 주입한다. 기본적으로 낮은 농도(7×10^{14})의 N형 채널을 만들고, 소스/드레인 접합부에는 높은 농도(3×10^{18})의 가우시안 분포 도핑을 주어 오믹 접촉이 잘 일어나게 한다.

```
#### MATERIAL/PHYSICAL MODELS ####
# Set parameters for amorphus silicon
MATERIAL region=2 mun=20 mup=1.5 nc300=2.5e20 \
nv300=2.5e20 eg300=1.9
```

MATERIAL: 비정질 실리콘의 전자 이동도($\mu_n=20$), 홀 이동도($\mu_p=1.5$), 유효 상태 밀도(n_{c300}) 등을 설정한다. 일반 단결정 실리콘보다 이동도가 훨씬 낮게 설정된 것이 특징이다.

```
# Defect states
DEFECTS nta=1e21 ntd=1e21 wta=0.033 wtd=0.049 \
nga=1.5e15 ngd=1.5e15 ega=0.62 egd=0.78 wga=0.15 wgd=0.15 \
sigtae=1e-17 sigtah=1e-15 sigtde=1e-15 sigtdh=1e-17 \
siggae=2e-16 siggah=2e-15 siggde=2e-15 siggdh=2e-16 continuous
```

비정질 반도체는 내부에 원자 배열이 불규칙해서 전자를 방해하는 함정이 많다. nta, ntd 같은 명령어는 이 함정들이 어디에, 얼마나 많은지를 컴퓨터에 알려주어 실제와 똑같이 시뮬레이션하게 한다.

```
# Gate workfunction
CONTACT number=1 aluminum
```

```
# Models
MODELS srh bgn temperature=300
```

```
# Impact ionization
IMPACT selberherr
```

```
#### NUMERICAL METHODS ####
METHOD itlimit=30 bicgst maxtrap=6 trap
```

```
#### SOLUTION SOLVING ####
SOLVE initial
SOLVE previous
SAVE outf=${name}_VD_inital.str
```

```
# Id-Vg calculations
# ramp the drain to 0.1V
```

```
SOLVE vdrain=0.025
SOLVE vdrain=0.05
SOLVE vdrain=0.1
# Ramp the gate and store the results
LOG outfile=${name}_VD_idvg.log
SOLVE vstep=0.25 vfinal=1.5 name=gate
SAVE outf=${name}_VD_vg1p5.str
SOLVE vgate=2.0 vstep=0.5 vfinal=20 name=gate
LOG off
```

실험 A: 이 수도꼭지는 얼마나 잘 작동하나?

1. 드레인에 0.1V라는 낮은 전압을 살짝 걸어준다.
2. 게이트 전압을 0V에서 20V까지 서서히 올린다.
3. 게이트 전압이 높아짐에 따라 소스와 드레인 사이에 전류가 급격히 흐르기 시작하는 지점을 찾는다.

```
# Id-Vd calculations
# Load structure file Vd=0.1V and Vg=1.5V
LOAD infile=${name}_VD_vg1p5.str master
SOLVE previous
# Ramp the drain and store the results
LOG outfile=${name}_VD_idvd.log master
SOLVE vdrain=0.1 vstep=0.1 vfinal=1 name=drain
SOLVE vdrain=1.25 vstep=0.25 vfinal=3 name=drain
SOLVE vdrain=3.25 vstep=0.25 vfinal=4 name=drain
```

실험 B: 전압을 세게 걸면 전류가 얼마나 나오나?

게이트 전압을 1.5V로 고정한다.

드레인 전압을 0V에서 4V까지 쭉 올린다.

압력이 세질 때 전류가 비례해서 늘어나다가 어느 순간 일정해지는 성능을 확인한다.

```
#### VISUALIZATION ####
victoryvisual ${name}_VD_initail.str:${name}_0.set \
  ${name}_VD_vg1p5.str:${name}_3.set \
  ${name}_VD_idvg.log \
  ${name}_VD_idvd.log

QUIT
```

